
Аннотация
Продукт масштаба «1С:Университет ПРОФ» при внедрении не обойдется без подстройки под особенности конкретного вуза. 

Вместе с новым функционалом приходят проблемы дальнейшего сопровождения, установки новых версий. Применение хорошо 
зарекомендовавших практик доработки не всегда решает задачу, потому что изменить типовой код без снятия конфигурации 
с поддержки нельзя, а это приведет к невозможности полностью автоматического обновления. Наиболее трудно сращивать 
изменения после серьезного рефакторинга кода со стороны поставщика. Если в вузе нет собственного специалиста, это может 
стать причиной отказа от доработок и существенного замедления внедрения.

В статье рассматривается способ доработки функционала конфигурации без потери возможности автоматических обнов-
лений. Этот способ подходит для всех современных решений «1С», включая используемые в образовательной деятельности: 
«1С:Университет» и «1С:Университет ПРОФ», «1С:Колледж», «1С:Общеобразовательное учреждение», «1С:Управление учебным 
центром». Приведена информация о развитии возможностей механизма расширения конфигурации, даны примеры встречаю-
щихся в вузах доработок, которые можно реализовать с использованием этого механизма. Приведенные примеры основаны на 
опыте реализации проектов автоматизации вузов Уральского федерального округа.

конфигурации с поддержки нельзя, а это приведет 
к невозможности полностью автоматического об-
новления. Наиболее трудно сращивать изменения 
после серьезного рефакторинга кода со стороны по-
ставщика. Если в вузе нет собственного специали-
ста, это может стать причиной отказа от доработок 
и существенного замедления внедрения. Тем не менее 
есть несколько способов доработки функционала 
«1С:Университет ПРОФ» без потери возможности 
автоматических обновлений.

Современные продукты фирмы «1С» содержат 
библиотеку стандартных подсистем (БСП), которая 
при подключении ко всем объектам конфигурации 
позволяет опытному пользователю самому поправить 
печатную форму, подключить внешнюю обработку, 
отчет. В «1С:Университет ПРОФ» встроена БСП, но 
решение об использовании и интеграции должна при-
нимать внедряющая организация [3]. Без изменения 
конфигурации можно добавить отчеты, но нельзя 
существенно изменить печатную форму документов 
или алгоритмы ввода данных.

1. Актуальность проблемы

Данные справочника «Внедренные решения» [1] 
позволяют утверждать, что треть учетных продуктов 
«1С» не требовали доработки конфигурации при 
вводе в эксплуатацию. Это говорит о соответствии 
их действующим нормативным документам.

С другой стороны, продукт масштаба «1С: Универ-
ситет ПРОФ», как и другое ERP-решение, не обой-
дется без подстройки под особенности конкретного 
вуза [2]. Наш опыт и опыт коллег показывают, что до 
80 % всех первоначальных доработок функционала 
связаны с изменениями печатных форм документов 
и отчетов. Когда справочники актуальны и данные 
сверены, появляется возможность решать более 
сложные управленческие и интеграционные задачи.

Вместе с новым функционалом приходят пробле-
мы дальнейшего сопровождения, установки новых 
версий. Применение хорошо зарекомендовавших 
практик доработки решает проблему частично — 
как ни старайся, а изменить типовой код без снятия 

2. Механизм расширений конфигураций

Помочь справиться с проблемой доработки функ-
ционала без снятия конфигурации с поддержки при-
зван механизм расширений конфигураций [4], при-
думанный как механизм кастомизации (изменения 
для нужд конкретного пользователя), который бы 
позволял:

•	 легко заменять кастомизированное решение 
на новую версию, избегая ручной работы по 
объединению конфигураций;

•	 включать кастомизацию при определенных 
условиях (например, если мы работаем в кон-
тексте определенной организации);

•	 снижать вероятность потери работоспособности 
кастомизации при переходе на новую версию 
исходной конфигурации;

•	 предоставлять возможность в случае проблем 
отключать кастомизацию для сохранения ра-
ботоспособности приложения.

Механизм расширений конфигураций подходит 
для всех современных решений «1С», включая 

Таблица

Развитие возможностей в расширениях конфигураций

№ 
п/п

Основные возможности, 
доступные в расширениях конфигураций

Версия технологической платформы

8.3.6 8.3.7 8.3.8 8.3.9 8.3.10 8.3.11 8.3.12 8.3.13

1 Изменять управляемые формы, существующие 
в типовой конфигурации + + + + + + + +

2 Добавлять новые подсистемы, изменять состав 
имеющихся подсистем + + + + + + + +

3 Изменять имеющиеся роли, добавляя объек-
ты, созданные в расширении + + + + + + + +

4 Изменять командный интерфейс (основного 
раздела, подсистем) + + + + + + + +

5 Добавлять новые отчеты и обработки + + + + + + + +

6 Создавать WS-ссылки, XDTO-пакеты, веб-
сервисы и HTTP-сервисы + + + + + + +

7 Публиковать интернет-сервисы расширения 
на веб-сервере + + + + + + +

8 Добавлять и изменять реквизиты и свойства 
заимствованных объектов, описывающих ос-
новные и вспомогательные формы

+ + + + + +

9 Заимствовать модули объектов, вызывать лю-
бые методы расширяемого модуля, добавлять 
обработчики событий

+ + + + +

10 Создавать новые роли и изменять права до-
ступа в заимствованных ролях + + + + +

11 Изменять и расширять макеты табличных 
документов + + + +

12 Добавлять реквизиты и создавать собственные 
справочники, документы, планы обмена и ре-
гистры сведений

+ + +

13 Формировать движения по регистрам для 
собственных и заимствованных документов + +

14 Создавать собственные перечисления + +

15 Использовать расширения в распределенной 
информационной базе + +

16 Создавать пользовательские стили и элементы 
стиля, заставку, логотип +

17 Создавать планы видов характеристик, планы 
счетов, планы видов расчета, регистры нако-
пления, бухгалтерии, расчета

+

18 Включать регистры любого вида в состав 
движений собственных и заимствованных 
документов расширения

+

используемые в образовательной деятельности 
[5–7]: «1С: Университет» и «1С: Университет ПРОФ», 
«1C:Библиотека», «1С:Колледж», «1С: Общеобразо-
вательное учреждение», «1С:  Управление учебным 
центром».

Расширение очень похоже на обычную конфи-
гурацию: оно так же представляется в виде дерева 
объектов, для работы с расширением используются 
те же приемы, что и для работы с обычной конфи-
гурацией. Но в отличие от конфигурации информа-
ционной базы расширение не содержит и не должно 
содержать всю дополнительную функциональность.

С момента выхода в 2015 году механизм расши-
рений подвергался переосмысливанию, доработкам 
и пересмотру функционала (см. табл.). Сейчас кон-
цепция механизма устоялась и ее можно применять 
без особого страха и рисков [8].

Процесс разработки расширения принципиально 
не отличается от обычного конфигурирования — 
расширение наполняется объектами конфигурации: 
заимствованными из конфигурации или созданными 
в расширении.

Чтобы редактировать имеющийся объект кон-
фигурации, необходимо воспользоваться командой 
Добавить в расширение контекстного меню объекта 
конфигурации. После выполнения этой команды 
в расширении создается заимствованный объект. 
При этом автоматически заимствуются также все ро-
дительские объекты, если они еще не заимствованы 
в данном расширении. Заимствовать объекты нужно:

•	 когда они необходимы для визуального кон-
струирования. Например, вы расширяете 
форму и добавляете реквизит формы типа 
СправочникСсылка.Валюты. Тогда вы должны 
заимствовать справочник «Валюты», чтобы 
при подключении к типовой конфигурации 
быть уверенными в том, что в ней такой спра-
вочник все еще есть;

•	 когда они необходимы для работы кода. На-
пример, в коде расширения вы обращаетесь 
к реквизиту Тип документа «Учебный план». 
Тогда этот реквизит также надо заимствовать, 
чтобы при подключении быть уверенным, что 
в типовой конфигурации у документа все еще 
существует такой реквизит.

При разработке расширений следует исходить из 
следующих предположений:

•	 расширение должно разрабатываться как авто-
номный продукт, не опирающийся на наличие 
или отсутствие других расширений;

•	 одновременно может быть подключено более 
одного расширения, которое расширяет один 
и тот же объект расширяемой конфигурации;

•	 порядок исполнения программных модулей 
в результирующей конфигурации опреде-
ляется порядком регистрации расширений 
в информационной базе расширяемой системы. 
Первым используется расширение, которое 
зарегистрировано последним;

•	 необработанное исключение, возникающее 
в любом из расширений (или расширяемой 
конфигурации), прерывает исполнение всей 
цепочки методов расширений и распростра-
няется в расширяемой конфигурации.

Можно выгрузить расширение в файл (с расши-
рением .cfe) и загрузить из файла. Форма управле-
ния расширениями конфигурации открывается по 
одноименной команде меню Все функции (рис. 1).

Разработка расширений конфигурации для 
«1С:Университет ПРОФ» при разработке практиче-
ских примеров использования рассматривается нами 
в следующем программном окружении:

•	 ОС Astra Linux Common Edition (2.12.8);
•	 СУБД PostgreSQL (9.6.6-1.1С);
•	 «1С:Предприятие 8.3» (8.3.12.1790);
•	 «1С:Университет ПРОФ, ред. 2.1» (2.1.4.3).

Рис. 1. Управление расширениями конфигурации

3. Примеры доработок с использованием 
расширений конфигураций

Приведем примеры встречающихся практически 
во всех вузах доработок, которые можно реализовать 
с использованием расширений конфигураций.

3.1. Изменение представлений объектов
При контроле правильности настроек программы, 

заполнения приказов и других документов удобно 
использовать короткие и информативные представ-
ления объектов информационной системы [9]. Согла-
ситесь, в заявлении абитуриента быстрее прочитыва-
ется «ПК (магистратура) 2018 г.» вместо «Приемная 
кампания 000000004 от 01.06.2018 0:00:00».

Обычно в «1С» ссылки на элементы справочников 
отображаются в виде кода или наименования (как 
задано в конфигураторе), а для документа основное 
представление — тип документа, номер, дата и вре-
мя. Изменить стандартное поведение разработчику 
позволяет процедура (обработчик события) платфор-
мы ОбработкаПолученияПредставления.

Предположим, что мы выполнили анализ типо-
вых ошибок при подготовке приказов (выбираются 
учебные планы вместо рабочих, не тот профиль 
и т. д.) и решили отображать в приказах поле с пла-
ном в виде «РП 2018 г. 07.03.01/АП очная № 335 
от 15.06.2015». Создадим новое расширение, которое 
будет использовать следующую возможность:

•	 заимствовать модули объектов, вызывать лю-
бые методы расширяемого модуля, добавлять 
обработчики событий (режим совместимости 
с версией 8.3.9).

Для формирования строки в нужном формате 
потребуется заимствовать в расширение следующие 
объекты основной конфигурации:

•	 документ «Учебный план» и реквизиты Тип, 
Специальность, УчебныйГод, ФормаОбучения;

•	 предопределенные значения справочника 
«Типы учебных планов».

При заимствовании реквизитов в расширении авто-
матически появляются соответ ствующие справочники. 
Желательно также добавить в расширение и поставить 
на контроль реквизиты автоматически заимствован-
ных справочников, которые мы будем использовать:

•	 справочник «Специальности» — реквизит 
КодСпециальности;

•	 справочник «Учебные годы» — реквизит На-
чальныйГод.

Реализуем в модуле менеджера документа «Учеб-
ный план» формирование строки представления 
в требуемом формате (листинг 1) и проверим работу 
алгоритма, открыв любой приказ (рис. 2). При вы-
боре типа расширяющих методов разработчику рас-
ширения, помимо конкретной прикладной задачи, 
нужно учитывать, что расширение разрабатывается 
с целью добавить свою функциональность, а не 
навсегда отказаться от того, что есть сейчас и что 
появится в дальнейшем в конфигурации. Поэтому 
обработчики событий в расширении рекомендуют 
определять с аннотациями &Перед и &После, чтобы 
расширяемый метод выполнился в любом случае [4].

Отметим особенности и ограничения формирова-
ния представления:

•	 для формирования представления доступны 
не все поля объекта, а только перечисленные 
в процедуре ОбработкаПолученияПолейПред
ставления;

•	 обращение к реквизитам через точку 
(Специальность.КодСпециальности) приведет 
к считыванию и кешированию всех полей 
и табличных частей, для сложных справочников 
лучше выбрать нужные поля запросом;

•	 в некоторых случаях необходимые для формиро-
вания представления поля содержатся в таблич-
ной части (например, профиль учебного плана) 

&После("ОбработкаПолученияПолейПредставления") 
Процедура Расш1_ОбработкаПолученияПолейПредставления(Поля, СтандартнаяОбработка) 
 Поля.Добавить("Специальность"); 
 Поля.Добавить("УчебныйГод"); 
 Поля.Добавить("ФормаОбучения"); 
КонецПроцедуры 
 
&После("ОбработкаПолученияПредставления") 
Процедура Расш1_ОбработкаПолученияПредставления(Данные, Представление, СтандартнаяОбработка) 
 СтандартнаяОбработка = Ложь; 
 Если Данные.Тип = Справочники.ТипыУчебныхПланов.УчебныйПлан Тогда 
  Префикс = "УП"; 
 ИначеЕсли Данные.Тип = Справочники.ТипыУчебныхПланов.РабочийПлан Тогда 
  Префикс = "РП"; 
 КонецЕсли; 
 Представление = СтроковыеФункцииКлиентСервер.ПодставитьПараметрыВСтроку( 
  "%1 %2 %3 %4 №%5 от %6", 
  Префикс, 
  Формат(Данные.УчебныйГод.НачальныйГод,"ДФ='yyyy ''г.'''"), 
  Данные.Специальность.КодСпециальности, 
  НРег(Данные.ФормаОбучения), 
  ПрефиксацияОбъектовКлиентСервер.НомерНаПечать(Данные.Номер), 
  Формат(Данные.Дата,"ДФ=dd.MM.yyyy") 
 ); 
КонецПроцедуры

Листинг 1. Процедуры менеджера документа «Учебный план»

и недоступны в параметре Данные. Можно до-
бавить реквизит Ссылка в массив необходимых 
полей и извлечь данные из базы запросом.

Опытные пользователи и разработчики, знакомые 
с «1С:Университет ПРОФ», отметят, что предложен-
ное решение не везде поменяет формат ссылок на учеб-
ные планы. Действительно, в некоторых отчетах, до-
говорах на обучение ссылка будет выглядеть примерно 
так: «Очная, факультет Архитектуры, Архитектура, 
Бакалавр, 2018–2019». В этих документах использу-
ется не сформированное платформой представление, 
а заполняемый при сохранении учебного или рабочего 
плана реквизит ПредставлениеУчебногоПлана.

При необходимости можно заимствовать мо-
дуль УчебныйПланСервер и переопределить в нем 
функцию ПолучитьПредставлениеУчебногоПлана 

(УчебныйПлан). В примере (листинг 2) используется 
аннотация &Вместо, так как &Перед и &После для 
функций недоступны. Если нужен результат рабо-
ты оригинальной функции, следует использовать 
специальную конструкцию ПродолжитьВызов(…) 
с исходными параметрами.

Для заполнения реквизита по новому алгоритму все 
учебные планы нужно будет однократно перепровести.

3.2. Изменение интерфейса и подсистем
Кафедры и деканаты часто жалуются на слишком 

большое количество элементов интерфейса. Действи-
тельно, пользователям лучше показывать ссылки 
только на используемые ими объекты — справоч-
ники, приказы, отчеты. Нужного результата можно 
достичь несколькими способами и без использования 
механизма расширений:

•	 настроить профили групп доступа, исходя из 
выполняемых трудовых функций: если поль-
зователю не дан ролевой доступ к какому-то 
объекту конфигурации, в интерфейсе тот по-
казан не будет [10];

•	 при использовании интерфейса «Такси» 
вручную убрать из навигации лишние ссыл-
ки, нажав на «шестеренку» (рис. 3), и затем 
скопировать настройки тем пользователям, 
которым нужен ограниченный набор команд.

Рис. 3. Возможность настройки интерфейса 
стандартными средствами

Когда использование расширений оправдано? 
В том случае, если при построении интерфейса необ-
ходимо переопределить логику разработчиков. Пере-
численные выше способы не позволят в подсистему 
«Управление студенческим составом» добавить отчет 
«Задолжники по договорам».

Предположим, в «1С:Университет ПРОФ» ра-
ботают сотрудники бухгалтерии, ответственные за 
взаиморасчеты со студентами, и нужно организовать 
простой интерфейс. Создадим новое расширение, 
которое будет использовать следующие возможности:

•	 добавлять новые подсистемы, изменять состав 
имеющихся подсистем;

•	 изменять имеющиеся роли, добавляя объекты, 
созданные в расширении;

•	 изменять командный интерфейс (основного 
раздела, подсистем);

•	 добавлять новые отчеты и обработки.
Продемонстрируем подход к решению постав-

ленной задачи:
•	 создадим структуру подсистем, исходя из на-

правлений деятельности вуза;
•	 заимствуем из основной конфигурации и вклю-

чим в новые подсистемы необходимые бухгал-
терии объекты — справочники, документы, 
отчеты;

&Вместо("ПолучитьПредставлениеУчебногоПлана") 

Функция Расш1_ПолучитьПредставлениеУчебногоПлана(УчебныйПлан) 

 Результат = ПродолжитьВызов(УчебныйПлан); 

 Возврат Результат; 

КонецФункции

Листинг 2. Переопределение функции общего модуля УчебныйПланСервер

Рис. 2. Представление ссылки на учебный план в приказе

•	 добавим новую команду Приказы на стипен-
дию, открывающую список приказов с заранее 
настроенным отбором;

•	 добавим новый отчет, который нужен бухгал-
терии;

•	 добавим роль, дающую доступ к новым объ-
ектам.

Убедимся, что в пользовательском режиме по-
явилась новая подсистема с требуемыми ссылками 
(рис. 4).

Таким же образом можно создать интерфейсы 
для других подразделений — деканатов, кафедр, 
военно-учетного стола и т. д. [11].

3.3. Устранение проблем совместной работы
Распространена ситуация, когда пользователь 

закрывает любую экранную форму нажатием на са-
мую большую кнопку — Записать и закрыть или 
Провести и закрыть.

Если документ только посмотрели (рис. 5) и ни-
чего не поменяли, команда Провести и закрыть 
запустит нежелательную цепочку действий — пере-
заполнение регистров, регистрацию изменений для 
обмена и т. д. Неприятнее другая ситуация — один 
пользователь редактирует данные документа или 

справочника, а другой в это время открыл тот же 
объект, посмотрел и закрыл командой Записать 
и закрыть. Первый увидит информацию об ошибке: 
«Данные были изменены или удалены другим поль-
зователем», не сможет сохранить изменения и будет 
долго ругаться. Сам того не желая, пользователь 
организовал ситуацию одновременного редактиро-
вания, которая должна специальным образом обра-
батываться как платформой, так и разработчиками 
решения [12].

Опытные пользователи скажут (и будут пра-
вы), что при попытке одновременного редакти-
рования одного и того же элемента справочника 
платформа сама заблокирует объект и выдаст 
сообщение об ошибке непосредственно в момент 
начала изменений. К сожалению, этот механизм 
не срабатывает при изменении поля ввода, кото-
рое непосредственно не связано с реквизитами 
редактируемого объекта. В реальных внедрениях 
встречаются, например, такие проблемы одновре-
менного редактирования:

•	 не блокируется элемент справочника «Физи-
ческие лица», если в форме студента редак-
тируются контактные данные и другие под-
чиненные данные;

Рис. 4. Интерфейс расширения в пользовательском режиме

Рис. 5. Активизированная по умолчанию кнопка документа

&НаКлиенте 

Процедура Расш3_ПередЗаписьюПеред(Отказ, ПараметрыЗаписи) 

 Если Не ЭтаФорма.Модифицированность Тогда 

  Отказ = Истина; 

  ЭтаФорма.Закрыть(); 

 КонецЕсли; 

КонецПроцедуры

Листинг 3. Проверка модифицированности перед записью объекта

Рис. 6. Добавление расширения обработчика события ПередЗаписью

•	 не блокируется при редактировании документ 
«Приказ».

И этот список легко можно продолжить.
Первая часть предлагаемого решения пробле-

мы — не выполнять стандартные операции, если 
данные на форме не менялись. Создадим новое 
расширение и заимствуем из формы документа 
«Учебный план» обработчик события ПередЗаписью 
(рис. 6), реализуем алгоритм доработки (листинг 3).

С одной стороны, подобные изменения необхо-
димо внести во многие экранные формы. С другой 
стороны, велика вероятность, что будущие измене-
ния основной конфигурации не отразятся на рабо-
тоспособности расширения.

Вторая часть предлагаемого решения — про-
граммно блокировать объект при начале редакти-
рования подчиненных данных, для этого можно 
использовать метод управляемой формы Заблоки-
роватьДанныеФормыДляРедактирования(). Форма 
элемента справочника «Физические лица» содержит 
27 динамических списков, для примера рассмотрим 
раздел «Контактная информация» (рис. 7). В рас-
ширение конфигурации заимствовать нужно форму 
записи регистра КонтактнаяИнформация.

Владельцем открытой формы редактирования 
контактных данных является таблица «Контакты» 
формы физического лица, данные которого нужно 
заблокировать для редактирования. Блокировку 
можно выполнить (листинг 4) как после открытия 
контактной информации, так и при изменении лю-
бого из реквизитов.

Устранение проблем совместной работы — от-
дельное направление в развитии прикладного 
решения, часто требующее глубокого анализа и су-
щественных изменений в программном коде [13]. 
Причины некоторых проблем можно самостоятельно 
устранить в расширении конфигурации.

3.4. Ограничение видимости персональных 
данных

Необходимость изменения форм ввода данных 
возникает в разных случаях. С помощью расширения 
можно доработать внешний вид формы, добавить 
в нее недостающие данные, дополнить ее команд-
ный интерфейс. Если же доработать модуль формы, 
можно не добавить, а наоборот, удалить или скрыть 
часть полей ввода.

В зависимости от должностных обязанностей 
пользователя в карточке физического лица можно 
не показывать часть вкладок и элементов формы. 
Например, ответственному за движение контингента 
не всегда нужно видеть паспортные данные, а инфор-
мацию о военной службе можно показывать только 
работникам военно-учетного стола [14].

Создадим новое расширение и продемонстрируем 
подход к решению задачи:

•	 заимствуем справочник «Физические лица» 
и форму элемента;

•	 создадим роли, исходя из выполняемых поль-
зователями обязанностей (рис. 8);

Рис. 8. Структура расширения для организации ролевой 
видимости данных

&НаКлиенте 

Процедура Расш3_ПриОткрытииПосле(Отказ) 

 УправляемаяФорма = ВладелецФормы; 

 Пока ТипЗнч(УправляемаяФорма) <> Тип("УправляемаяФорма") Цикл 

  УправляемаяФорма = УправляемаяФорма.Родитель; 

 КонецЦикла; 

 УправляемаяФорма.ЗаблокироватьДанныеФормыДляРедактирования(); 

КонецПроцедуры

Листинг 4. Блокировка объекта для изменения при начале редактирования

Рис. 7. Исходные данные для списка «Контакты» физического лица

•	 реализуем ограничение видимости при созда-
нии формы (листинг 5);

•	 в пользовательском режиме в разделе «Адми-
нистрирование», «Настройки пользователей 
и прав» создадим профиль групп доступа 
(рис. 9).

Рис. 9. Создание профиля доступа с использованием роли 
из расширения

Предлагаемое решение позволяет организовать 
динамическое управление видимостью элементов 
формы и не дает возможности пользователю вклю-
чить самостоятельно скрытые элементы с использо-
ванием команды Изменить форму.

3.5. Предотвращение появления 
дублей в справочниках

Как уже было отмечено, причинами многих про-
блем, возникающих при эксплуатации системы, явля-
ется неверный ввод исходных данных пользователями. 
Например, операторы приемной комиссии заполняют 

десятками дублирующих записей справочник «Контр-
агенты», делают ошибки и опечатки в наименованиях. 
Перед сохранением введенных оператором данных 
новой организации необходимо проверить, что такие 
же или похожие записи в справочнике отсутствуют.

Для поиска и удаления дублей удобно использо-
вать одноименную подсистему библиотеки стандарт-
ных подсистем, ключевым преимуществом которой 
является поиск по подобию (нечеткий поиск, fuzzy 
search) [15].

Создадим новое расширение и продемонстрируем 
подход к решению задачи:

•	 заимствуем общий модуль ПоискИУдаление-
ДублейПереопределяемый и добавим справоч-
ник «Контрагенты» в метод ПриОпределении-
ОбъектовСПоискомДублей();

•	 заимствуем справочник «Контрагенты» и фор-
му элемента;

•	 в менеджере справочника опишем параметры 
поиска дублей — совпадение по типу и подобие 
наименования, а если необходимо, то и реа-
лизуем свой собственный алгоритм сравнения 
при поиске дублей (листинг 6);

•	 в форме элемента добавим обработчик события 
«Перед записью», в нем обратимся к методу 
БСП НайтиДублиЭлемента() и просто вы-
ведем полученный список возможных дублей 
на экран (листинг 7).

Проверим наше решение — создадим новый эле-
мент с опечаткой в наименовании (интитут) и по-
пробуем его сохранить — будут показаны найденные 
дубликаты (рис. 10).

&НаСервере 

Процедура Расш4_ПриСозданииНаСервереПосле(Отказ, СтандартнаяОбработка) 

 Если НЕ Пользователи.РолиДоступны("Расш4_ВУС") Тогда 

  Элементы.ВоеннаяСлужба.Видимость = Ложь; 

 КонецЕсли; 

КонецПроцедуры

Листинг 5. Ограничение видимости элементов формы на основе ролей

Процедура ПараметрыПоискаДублей(ПрикладныеПараметры, ДополнительныеПараметры) Экспорт 

 ДобавитьПравило(ПрикладныеПараметры,"Равно","ТипКонтрагента"); 

 ДобавитьПравило(ПрикладныеПараметры,"Подобно","Наименование"); 

КонецПроцедуры 

 

Процедура ДобавитьПравило(ПрикладныеПараметры,Правило,Реквизит) 

 Отбор = Новый Структура("Правило,Реквизит",Правило,Реквизит); 

 Строки = ПрикладныеПараметры.ПравилаПоиска.НайтиСтроки(Отбор); 

 Если Строки.Количество() = 0 Тогда 

  НовоеПравило = ПрикладныеПараметры.ПравилаПоиска.Добавить(); 

  НовоеПравило.Правило = Правило; 

  НовоеПравило.Реквизит = Реквизит; 

 КонецЕсли; 

КонецПроцедуры 

 

Процедура ПриПоискеДублей(ТаблицаКандидатов, ДопПараметры) Экспорт 

 Для Каждого Строка Из ТаблицаКандидатов Цикл 

  Строка.ЭтоДубли = Истина; // анализ :) 

 КонецЦикла; 

КонецПроцедуры

Листинг 6. Настройки поиска в модуле менеджера справочника «Контрагенты»

3.6. Доработка печатных форм

Изменения в печатных формах — обязательный 
этап внедрения «1С:Университет». Заказчиками до-
работок обычно являются приемная комиссия и под-
разделения, ответственные за подготовку приказов 
и справок. Изменения кода конфигурации в ряде 
случаев можно избежать:

•	 если печатная форма реализована с использо-
ванием БСП (тогда ее может самостоятельно 
редактировать ответственный пользователь) 
(рис. 11);

•	 если печатная форма реализуется как внешняя 
обработка;

•	 если настроена бесшовная интеграция с си-
стемой электронного документооборота 
«1С:Документооборот», в которой подготов-
лены шаблоны документов [16].

Если перечисленные варианты не подходят или 
требуется не разработка нового макета, а изменение 
уже существующего, можно воспользоваться воз-
можностями расширений. Наши коллеги, например, 
отмечают, что при всей гибкости настройки печати 

приказов «1С:Университет» не дает возможности 
глобально поменять шапку документов [17].

Создадим новое расширение, которое будет ис-
пользовать такой функционал:

•	 изменять и расширять макеты табличных до-
кументов (появилось в версии 8.3.10)

Из основной конфигурации заимствуем общий 
макет ШапкаВуза и настроим печатную форму 
(рис. 12).

Обратим внимание на особенность заимствования 
макетов табличных документов: можно настраивать 
способ расширения — объединение или замещение 
(рис. 13). В первом случае (это настройка по умол-
чанию) построение итогового документа будет про-
исходить немного медленнее.

Команды печати могут быть добавлены в рас-
ширение различными способами [10].

3.7. Добавление веб-сервиса
Информирование абитуриентов о плане набора 

и количестве поданных заявлений обычно делается 
через сайт вуза. Однако реализация на стороне сер-
вера «1С» REST-интерфейса и использование веб-

&НаКлиенте 

Процедура Расш5_ПередЗаписьюПосле(Отказ, ПараметрыЗаписи) 

 СписокДублей = Расш5_НайтиДублиЭлемента(Объект); 

 Если СписокДублей.Количество() > 0 Тогда 

  Отказ = Истина; 

  СписокДублей.ПоказатьВыборЭлемента("Похожие элементы"); 

 КонецЕсли; 

КонецПроцедуры 

 

&НаСервереБезКонтекста 

Функция Расш5_НайтиДублиЭлемента(Знач Проверяемый) 

 Результат = Новый СписокЗначений; 

 ЭталонныйОбъект = ДанныеФормыВЗначение(Проверяемый,Тип("СправочникОбъект.Контрагенты")); 

 ВозможныеДубли = ПоискИУдалениеДублей.НайтиДублиЭлемента( 

  "Справочник.Контрагенты",ЭталонныйОбъект,Неопределено); 

 Для Каждого Строка Из ВозможныеДубли Цикл 

  Если ЭталонныйОбъект.Ссылка <> Строка.Ссылка Тогда 

   Результат.Добавить(Строка.Ссылка); 

  КонецЕсли; 

 КонецЦикла; 

 Возврат Результат; 

КонецФункции

Листинг 7. Поиск дублей перед записью элемента справочника «Контрагенты»

Рис. 10. Найденные дубли контрагента

приложений на основе Angular, React и пр. [18, 19] 
позволяют организовать информационную панель 
вообще без изменения серверного кода сайта вуза 
и настройки обменов данными.

Создадим новое расширение, использующее воз-
можности:

•	 создавать WS-ссылки, XDTO-пакеты, веб-
сервисы и HTTP-сервисы;

•	 публиковать интернет-сервисы расширения на 
веб-сервере.

Создадим в расширении веб-сервис, в качестве 
примера продемонстрируем получение и публика-
цию в формате JSON плана набора (рис. 14, ли-
стинг 8).

Выберем в конфигураторе команду Публикация 
на веб-сервере в меню Администрирование и про-

Рис. 11. Самостоятельное редактирование пользователем макета печатной формы

Рис. 12. Общий макет в расширении конфигурации

Рис. 13. Особенности расширения макетов табличных документов

Рис. 14. Структура веб-сервиса в расширении

Функция ШаблонПланНабораПолучить(Запрос) 

 Ответ = Новый HTTPСервисОтвет(200); 

 Ответ.Заголовки.Вставить("Content-type", "application/json"); 

 Запрос = Новый Запрос; 

 Запрос.УстановитьПараметр("ГодПоступления",'20180101000000'); 

 Запрос.Текст = 

 "ВЫБРАТЬ 

 | Планы.КонкурснаяГруппа.Представление КАК КонкурснаяГруппа, 

 | Планы.ОснованиеПоступления.Представление КАК Основание, 

 | Планы.УровеньБюджета.Представление КАК УровеньБюджета, 

 | Планы.КоличествоМест КАК Мест 

 |ИЗ 

 | РегистрСведений.ПланыНабора.СрезПоследних(, ) КАК Планы 

 |ГДЕ 

 | Планы.ПриемнаяКампания.ГодПоступления = &ГодПоступления"; 

 Выб = Запрос.Выполнить().Выбрать(); 

 мПланНабора = Новый Массив; 

 Пока Выб.Следующий() Цикл 

мПланНабора.Добавить(Новый Структура("КонкурснаяГруппа,Основание,УровеньБюджета,КоличествоМест", 

Выб.КонкурснаяГруппа, Выб.Основание, Выб.УровеньБюджета, Выб.Мест)); 

 КонецЦикла; 

 JSON = Новый ЗаписьJSON; 

 JSON.УстановитьСтроку(); ЗаписатьJSON(JSON,мПланНабора); 

 Ответ.УстановитьТелоИзСтроки(JSON.Закрыть()); 

 Возврат Ответ; 

КонецФункции

Листинг 8. Получение и публикация плана набора через веб-сервис

Рис. 15. Настройка публикации веб- и HTTP-сервисов расширений

верим, что HTTP-сервисы расширений публикуют-
ся (рис. 15). Затем проверим, что при обращении 
к веб-сервису данные возвращаются в нужном виде 
(листинг 9).

4. Выводы

Широко известное достоинство программных 
продуктов «1С» — открытый код и возможность 
кастомизации типовых конфигураций на платфор-
ме «1С:Предприятие 8» — при использовании рас-
ширений выходит на новый качественный уровень. 
Отлаженная на одном экземпляре информационной 
системы доработка может быть оформлена в виде 
отторгаемой надстройки и перенесена в аналогичную 
систему. 

Бывают ситуации, когда многим пользователям 
типового прикладного решения нужна одна и та же 
функциональность, но вносить ее в типовую конфи-
гурацию не имеет смысла, потому что алгоритмы 
сильно разнятся в зависимости от региона, учредите-
ля или местных инструкций. Не нужно требовать от 
разработчиков «1С:Университет ПРОФ» реализацию 
всех мыслимых и немыслимых особенностей учета. 
Необходимо обеспечить современную инфраструкту-
ру, в первую очередь, поддержку последней версии 
технологической платформы и стандартных библио-
тек (БСП, БИД и др.).

Отметим также, что само слово «расширение» 
заставляет воспринимать технологию как средство 
для модернизации типового функционала. Не нужно 
откидывать альтернативные идеи применения рас-
ширений. Одним из интересных способов использо-
вания расширений является создание «заплаток» для 
решения проблемных ситуаций или оперативного 
исправления ошибок конфигурации (см. рис. 1).

Фирма «1С» пока не спешит агитировать раз-
работчиков к созданию полноценных тиражных ре-
шений на основе механизма расширений, мотивируя 
это молодостью технологии и не совсем подходящей 
архитектурой. Несмотря на это, уже сейчас вполне 
реально применять технологию расширений для 
создания небольших универсальных плагинов или 
при работе в модели сервиса [20]. Мы убедились 
в том, что технология доработки конфигурации 
при помощи механизма расширений применима во 
многих случаях внедрения современных решений 
«1С», включая используемые в образовательной 
деятельности. 

Н О В О С Т И

стандарт содержанием. Это необходимо, чтобы на деле, 
а  не на словах в  России была реализована политика 
единого образовательного пространства. Без содержания 
в  стандарте мы не сможем никогда навести порядок 
с оценочными — промежуточными и итоговыми — про-
цедурами и с качеством обучения наших детей», — ска-
зала Татьяна Юрьевна.

Т. Ю. Синюгина также информировала о  работе, 
которую Минпросвещения России проведет совместно 
с регионами в целях обеспечения качества преподавания.

«В  рамках федерального проекта “Современная 
школа” мы поставили для себя достаточно амбициоз-
ные задачи: достижение качественного результата через 
оснащение не только городских, но и  сельских школ, 
которые впервые включены в  национальный проект 
и получат средства на развитие профилей гуманитарного, 
естественно-научного и  цифрового направления. Это 
будет происходить через подготовку и  переподготовку 
учителей, а также создание совершенно новых структур, 
которые позволяли бы учителю системно повышать свою 
квалификацию, понимая, какие потребности сегодня 
есть», — сообщила заместитель министра.

По итогам круглого стола его участники сформиро-
вали предложения, которые будут направлены в  Совет 
Федерации и Минпросвещения России.

11 апреля в  Совете Федерации Федерального Со-
брания Российской Федерации состоялся круглый стол 
на тему «О  законодательном обеспечении стандартов 
качества общего образования по предметам естественно-
научного цикла». В  обсуждении приняла участие заме-
ститель министра просвещения Российской Федерации 
Т. Ю. Синюгина.

Т. Ю. Синюгина отметила, что обновление ФГОС 
будет способствовать расширению профессионального 
творчества учителей.

«Важно отметить, что мы не стараемся что-то запре-
тить. Речь идет о профессиональном творчестве, свободе 
наших учителей, которые никем не будут ограничены. Мы 
выступаем за то, чтобы в стандарте были четко пропи-
саны требования к предметным результатам для каждой 
дисциплины и учитель, вне зависимости от места работы, 
понимал, что минимально необходимо на каждом этапе 
прохождения того или иного предмета»,  —  сообщила 
Татьяна Юрьевна.

Заместитель министра подчеркнула, что одна из 
главных целей работы ведомства  — создание единого 
образовательного пространства.

«Мы будем системно и  последовательно, опираясь 
на предложения, широкое обсуждение, поддержку со 
стороны профессионального сообщества, наполнять 

Образовательные стандарты не ограничат творческую свободу учителей

(По материалам, предоставленным пресс-службой Минпросвещения России)


